

# WINNIE THE POOH IN THE HUNDRED ACRE WOOD

## COMMODORE COMPUTER CHALLENGE ACTIVITY 1 — C64 Visual Map

Name \_\_\_\_\_ Date \_\_\_\_\_

There is a map on the screen. Where you are is represented by a red box, where you want to go is indicated by a yellow X. The computer challenges you to type in N, S, E or W to move North, South, East or West and get to the yellow X.

```
10 DIM I(13,2),N(13),S(13),E(13),W(13)
20 PRINT CHR$(147):C=3:T=102
30 FOR X=10 TO 28:Y=4:GOSUB 300:Y=13:GOSUB 300:NEXT X
40 FOR X=0 TO 28:Y=8:GOSUB 300:NEXT X
50 FOR Y=0 TO 13:FOR X=10 TO 28 STEP 9:GOSUB 300:NEXT X:NEXT Y
60 FOR J=1 TO 13:READ I(J,1),I(J,2),N(J),S(J),E(J),W(J):NEXT J
70 P=INT(RND(1)*13)+1:G=INT(RND(1)*13)+1:IF P=G THEN GOTO 70
80 C=2:X=I(P,1):Y=I(P,2):GOSUB 300
90 C=1:T=24:X=I(G,1):Y=I(G,2):GOSUB 300:T=102
95 PRINT""
96 FOR J=1 TO 16:PRINT:NEXT J
100 PRINT "YOU ARE THE RED BOX.":PRINT "X MARKS THE SPOT YOU WANT TO GET TO."
110 PRINT "TYPE N, S, E, OR W TO MOVE, THEN RETURN"
120 INPUT A$:R=0:IF A$="N" THEN R=N(P)
125 IF A$="W" THEN R=W(P)
130 IF A$="S" THEN R=S(P)
140 IF A$="E" THEN R=E(P)
150 IF R=0 THEN GOTO 400
160 C=3:X=I(P,1):Y=I(P,2):GOSUB 300
170 C=2:X=I(R,1):Y=I(R,2):GOSUB 300
180 P=R:IF P<>G THEN GOTO 95
190 PRINT"YOU ARRIVED! YOU ARE A GOOD MAP READER!"
200 RESTORE:FOR J=1 TO 3000:NEXT J:GOTO 20
210 DATA 10,4,10,4,2,0,19,4,11,5,3,1,28,4,12,6,0,2,10,8,1,7,5,13
220 DATA 19,8,2,8,6,4,28,8,3,9,0,5,10,13,4,0,8,0,19,13,5,0,9,7
230 DATA 28,13,6,0,0,8,10,0,0,1,0,0,19,0,0,2,0,0,28,0,0,3,0,0
240 DATA 0,8,0,0,4,0
300 POKE 55296+X+40*Y,C:POKE 1024+X+40*Y,T:RETURN
400 PRINT "YOU CANNOT GO THAT WAY";
405 FOR Z=1 TO 300:NEXT Z
410 FOR Z=1 TO 22
420 PRINT CHR$(20);
430 NEXT Z
440 PRINT CHR$(20);
450 PRINT:GOTO 95
```



# WINNIE THE POOH IN THE HUNDRED ACRE WOOD

## COMMODORE COMPUTER CHALLENGE ACTIVITY 2 — Blind Map

Name \_\_\_\_\_ Date \_\_\_\_\_

Pooh has lost his honey. The computer challenges you to help find it. Type in N, S, E or W to go North, South, East or West. The computer will tell you if you are getting closer or farther away.

```
10 W=INT(RND(1)*9)+2:H=INT(RND(1)*9)+2
20 X=5:Y=5:D=ABS(W-X):E=ABS(H-Y)
30 PRINT CHR$(147):PRINT"HELP POOH FIND HIS MISSING HONEY"
40 PRINT:PRINT"TYPE THE DIRECTION HE SHOULD WALK,":PRINT
50 PRINT" N, S, E, OR W, THEN HIT RETURN.":PRINT
60 PRINT
70 INPUT A$:IF A$="N" THEN Y=Y+1:GOTO 120
80 IF A$="S" THEN Y=Y-1:GOTO 120
90 IF A$="E" THEN X=X+1:GOTO 120
100 IF A$="W" THEN X=X-1:GOTO 120
110 PRINT "PLEASE ENTER A DIRECTION: N, S, E, W":GOTO 60
120 Z=D:P=E:D=ABS(W-X):E=ABS(H-Y)
130 IF D=0 AND E=0 THEN GOTO 170
140 IF (D=0 AND E=P) OR (E=0 AND D=Z) THEN PRINT "YOU ARE VERY CLOSE!":GOTO 60
150 IF D<Z OR E<P THEN PRINT "YOU ARE GETTING CLOSER!":GOTO 60
160 IF D>Z OR E>P THEN PRINT "YOU ARE GETTING FARTHER AWAY!":GOTO 60
170 PRINT"YOU FOUND IT!":FOR I=1 TO 3000:NEXT I:GOTO 10
```

